## Chapter 7 Concurrent Systems

Finally, we are going to employ the concepts introduced in the previous chapter for their core purpose: the modeling and analysis of systems with concurrently executing components. Examples of such systems are programs with multiple processes or threads that cooperate via shared variables, as well as multiple independent programs that interact by exchanging messages over a network.

The core idea is to replace the somewhat intricate notion of "concurrency" by the more tractable notion of "nondeterminism". While in a concurrent system multiple actions may be executed simultaneously, the nondeterministic model of such a system considers the execution of individual actions only, but of *any* possible action; thus the analysis of the model investigates all possible "interleavings" of actions. Although this does not consider the truly simultaneous execution of two actions  $a_1$  and  $a_2$ , one may argue that it suffices to consider just two possibilities, that  $a_1$  is executed before  $a_2$ , and that  $a_2$  is executed before  $a_1$ : either both actions affect a common physical component, then this component has to arrange them in one of these orders to achieve a well-defined effect; or they only affect components at different locations, then the notion of "simultaneity" is meaningless (due to the principle of relativity).

When reasoning about such models, we will focus on their *safety properties*, i.e., properties whose violation can be observed at a particular point of the system execution; such properties ensure that "nothing bad can ever happen". The verification of a safety property requires the formulation of a *system invariant* that constrains the set of reachable system states as closely as possible and that implies the safety property. The correctness of the invariant is shown by an *induction proof*: all initial system states must satisfy the invariant (the induction base), and, if the invariant holds in a state (the induction hypothesis), it must hold again after every possible action of the system (the induction step); typically the safety property of interest has to be be sufficiently strengthened to yield valid induction steps. Indeed the formulation of *inductive invariants* is the core problem of concurrent system verification; only via such invariants a system is truly understood.

## 7.3 A Resource Allocator

```
87 shared system AlternatingBitNetwork
 88 {
                                                                                                                                                 Execute operation
      // the messages sent and received (local to each process)
 89
                                                                                                                                                 Verify specification precondition
 90 var sent:Msg = Default;
91 var rcvd:Msg = Default;
                                                                                                                                                   👙 le index value legal?
                                                                                                                                                   🔅 le index value legal?
                                                                                                                                                   Is index value legal?
 93
        // the protocol bits
                                                                                                                                                   Is index value legal?
        var sbit:Bit = 1:
 94
 95
        var sack:Bit = 1;
                                                                                                                                                   A le index value legal?
 96
        var rbit:Bit = 1;
                                                                                                                                                   👙 is index value legal?
 97
                                                                                                                                                   Is index value legal?
 98
        // the communication channels
                                                                                                                                                   Is index value legal?
 99
       var msqg:Queue = (len:0,pack:Array[M,Package]((msg:Default,bit:1)));
var ackq:Queue = (len:0,pack:Array[M,Package]((msg:Default,bit:1)));
                                                                                                                                                   Is index value legal?
100
                                                                                                                                                   👙 is index value legal?
        // the history of the messages sent and received
                                                                                                                                                   Sindex value legal?
102
        var smsgs:Seq = Array[N,Msg](Default);
var snum:Index = 0;
                                                                                                                                                   Is index value legal?
103

    Verify specification

104
       var rmsgs:Seq = Array[N,Msg](Default);
var rnum:Index = 0:
105
                                                                                                                                                  Ones system invariant initially hold?
106
                                                                                                                                                   Does system invariant initially hold?
107
                                                                                                                                                  Does system invariant initially hold?
       // safety property: the messages received are the messages sent
invariant rnum ≤ snum ∧ ∀i:Index with i < rnum. rmsos[i] = smsos[i]:</pre>
108
                                                                                                                                                  Does system invariant initially hold?
109
                                                                                                                                                  Does system invariant initially hold?
116
                                                                                                                                                  Does system invariant initially hold?
        // invariants from the shared memory version
                                                                                                                                                  Does system invariant initially hold?
        invariant snum-1 ≤ rnum;
invariant sack = sbit ⇒ rnum = snum;
                                                                                                                                                  Does system invariant initially hold?
                                                                                                                                                  Does system invariant initially hold?
114
        invariant rbit ≠ sbit ∧ rnum < N → rnum < snum ∧ sent = smsgs[snum-1];
                                                                                                                                                  Does system invariant initially hold?

    action sender

116
        // additional invariants
117 invariant rnum < snum → rbit ≠ sbit;
118 invariant ∀i:N[M] with i < msgq.len.</pre>
                                                                                                                                                action resend
                                                                                                                                                action receiveAck
       invariant visup with i < msqq.ten;
(msqq.beck[i].bit + rbit = msqq.ten;
invariant visup with i=1 < msqq.ten;
(msqq.beck[i].bit + rbit = msqq.ten;
119
                                                                                                                                                action receiver
120
                                                                                                                                                action sendAck
      (msqu.pack[i].blt # folt = msqu.pack[i+1].blt # folt];
invariant ¥1:N[M] with i+1 < ack[.en.
(ackq.pack[i].bit # sack = ackq.pack[i+1].bit # sack);
invariant sack = sbit = (¥i:N[M] with i < ackq.len. (ackq.pack[i].bit = rbit));
invariant sack = rbit = (¥i:N[M] with i < ackq.len. (ackq.pack[i].bit = rbit));</pre>
                                                                                                                                                ▶ action loseMsg
                                                                                                                                                I action loseAck
                                                                                                                                                  Verify initialization precondtions
124

    Verify action preconditions
```

Fig. 7.7 Verification of the Alternating Bit Protocol (Distributed Version)

## 7.3 A Resource Allocator

We consider a system described in [44] which consists of a server and a set of clients (see Figure 7.8). The server manages a pool of resources which clients request from the server by sending corresponding messages; the server grants these requests by sending corresponding replies. The central safety property of the system is that the server grants every resource to at most one client at a time, i.e., no two clients may simultaneously use the same resource.

The problem becomes complex, because both requests and grants do not only refer to single resources; in particular, every client may request any set resources. However, the server may respond by a message that contains only some of the requested resources; if it does not immediately grant all resources, the server will send later further messages that grant more of them, until the complete request is satisfied. Furthermore, as soon as a client holds some of the requested resources, it may (even if its request has not yet been satisfied completely) return some of them to the server. However, a client may not send another request for new resources before it has received and returned all the resources from its previous request.

To fairly satisfy requests from the various clients, the server keeps track of the sequence of still pending requests in the order in which they were received. The server grants a resource to a client if there is no earlier request from another client for the same resource.



Fig. 7.8 A Resource Allocator

We will model this system in RISCAL based on the following declarations:

```
val C:\mathbb{N}; val R:\mathbb{N};
axiom notNull \Leftrightarrow C > 0 \wedge R > 0;
type Client = \mathbb{N}[C-1];
type Resource = \mathbb{N}[R-1];
type Position = \mathbb{N}[C];
```

Here the model parameter C > 0 denotes the number of clients while R > 0 denotes the number of resources. Clients are denoted by values of type *Client* while resources are denoted by values of type *Resource*. Values of type *Position* represent the positions of requests queued in the server; smaller values denote requests that have been received earlier and that therefore have higher priority (value *C* indicates "no position").

The model makes use of the following auxiliary operations:

```
fun position(c:Client,pos:Map[Client,Position]):Position =
    let p = pos[c] in
    if p < C then
        p
    else if ∀c0:Client. pos[c0] = C then
        0
    else
        1 + max c0:Client with pos[c0] < C. pos[c0];</pre>
```

The value of position(c, pos) denotes the position of client c in the "queue" of requests, according to the mapping *pos* of clients to positions (this queue is purely "virtual", i.e., it is only represented by *pos*). If this mapping already contains a position for c, we use this position; otherwise, if the mapping is empty, the position is 0 (the client is at the "front" of the queue); if the mapping is not empty, the position is one plus the maximum position in the mapping (the client is at the "back" of the queue).

```
proc remove(pos:Map[Client,Position], c:Client):Map[Client,Position]
ensures ∀c0:Client. result[c0] = if c0 = c then C else
if pos[c] < pos[c0] ∧ pos[c0] < C then pos[c0]-1 else pos[c0];
{
  var p:Map[Client,Position] = pos;
  for c0:Client with p[c] < p[c0] ∧ p[c0] < C do
    p[c0] := p[c0]-1;
  p[c] := C;
  return p;
}
```

The value of remove(pos, c) denotes the mapping of clients to positions that arises from *pos* if we remove client *c* from the queue of requests: the position of all clients before *c* remain unchanged while the positions of all clients after *c* are decreased by one; the position of *c* itself is set to *C* ("no position").

```
pred grant(c:Client,S:Set[Resource],unsat:Map[Client,Set[Resource]],
    alloc:Map[Client,Set[Resource]],pos:Map[Client,Position])
⇔ pos[c] < C ∧ S ≠ Ø[Resource] ∧
    (∀r∈S. r∈unsat[c] ∧ ¬∃c0:Client. r∈alloc[c0]) ∧
    (∀c0:Client with pos[c0]<pos[c]. ∀r∈S. r∉unsat[c0]);</pre>
```

The truth value of *grant*(*c*, *S*, *unsat*, *alloc*, *pos*) states whether the server may allocate to client *c* the resource set *S*, depending on its state described by the variables *unsat*, *alloc*, and *pos* which will be further explained below.

## A Shared Model

Our first model is that of a system with nondeterministic transitions operating on a set of shared variables; in particular, the state of the network isbdefined by the set of messages that are currently in transit. These messages are formed according to the following declarations:

```
type Tag = N[2];
val request = 0; val allocate = 1; val giveback = 2;
type Message = Record[tag:Tag,client:Client,resources:Set[Resource]];
```

Thus a message is a value *m* of type *Message* where the record field  $m.tag \in \{request, allocate, giveback\}$  denotes the purpose of the message, *m.client* denotes the client sending respectively receiving the message, and

*m.resources* denotes the set of resources carried by the message. On the basis of these declarations, the model can now be formalized as follows:

```
shared system SharedAllocator
{
  var unsat:Map[Client,Set[Resource]] =
    Map[Client,Set[Resource]](Ø[Resource]);
  var alloc:Map[Client,Set[Resource]] =
     Map[Client,Set[Resource]](Ø[Resource]);
  var pos:Map[Client,Position] = Map[Client,Position](C);
  var requests: Map[Client,Set[Resource]] =
    Map[Client,Set[Resource]](Ø[Resource]);
  var holding: Map[Client,Set[Resource]] =
    Map[Client,Set[Resource]](Ø[Resource]);
  var network: Set[Message] = \emptyset[Message];
  action serverRequest(m:Message) with
    m \in network \land m.tag = request \land alloc[m.client] = \varnothing[Resource];
  {
    network := network \{m\}; val c = m.client; val S = m.resources;
    unsat[c] := unsat[c]US; pos[c] := position(c,pos);
  }
  action serverAllocate(c:Client,S:Set[Resource]) with
    grant(c,S,unsat,alloc,pos);
  {
    alloc[c] := alloc[c]US; unsat[c] := unsat[c]\S;
    if unsat[c] = Ø[Resource] then pos := remove(pos,c);
    network := network\cup{\langletag:allocate,client:c,resources:S\rangle};
  }
  action serverGiveBack(m:Message) with m \in network \land m.tag = giveback;
  {
    network := network \{m\}; val c = m.client; val S = m.resources;
    alloc[c] \coloneqq alloc[c] \setminus S;
  }
  action clientRequest(c:Client,S:Set[Resource]) with
    requests[c]=\emptyset[Resource] \land holding[c]=\emptyset[Resource] \land S\neq \emptyset[Resource];
  Ł
    requests[c] := S;
    network := network\cup{\langletag:request,client:c,resources:S\rangle};
  }
  action clientAllocate(m:Message) with m \in network \land m.tag = allocate;
  ł
    network := network \{m\}; val c = m.client; val S = m.resources;
    requests[c] := requests[c]\S; holding[c] := holding[c]\cupS;
  }
  action clientGiveBack(c:Client,S:Set[Resource]) with
    S \neq \emptyset[Resource] \land S \subseteq holding[c];
  {
    holding[c] := holding[c] \setminus S;
    network := network\cup{\langletag:giveback,client:c,resources:S\rangle};
  }
}
```